

#T25

# Задача контроля остатков

## Методики контроля остатков. Понятие транзакции.

Профессия «Программист 1С»



# Из этой лекции вы узнаете

- Что такое **транзакция**
- Как выполнить определенный **алгоритм в транзакции**
- Какие **события** возникают при проведении/записи документа и как они связаны с транзакциями
- Как решать **задачу контроля остатков** («старая» и «новая» методика)

# **ЗАДАЧА КОНТРОЛЯ ОСТАТКОВ**

# Постановка задачи

При проведении документа «Расходная накладная» нужно проверять: достаточно ли товара на складе для отгрузки.

Если товара не хватает, то документ не должен проводиться + пользователь должен видеть каких товаров и в каком количестве не хватает.


# ЧТО ТАКОЕ ТРАНЗАКЦИЯ

# Транзакция

- логически связанная и **неделимая** последовательность действий
- транзакция может быть либо выполнена целиком, либо вообще не выполнена.

# Пример транзакции

- Перевод 100 руб. со счета «А» на счет «Б»



# Ключевые свойства транзакций (ACID)

- **Атомарность (Atomicity)**  
никакая транзакция не должна быть зафиксирована частично
- **Согласованность (Consistency)**  
система находится в согласованном состоянии до начала транзакции и должна остаться в согласованном состоянии после завершения транзакции
- **Изолированность (Isolation)**  
во время выполнения транзакции параллельные транзакции не должны оказывать влияние на её результат
- **Надежность (Durability)**  
в случае сбоя изменения, сделанные успешно завершённой транзакцией, должны остаться сохранёнными после возвращения системы в работу

# Какие бывают транзакции

- **Неявные**

- Открываются при выполнении любых действий, связанных с изменением информации, которая хранится в базе
  - *ДокОбъект.Записать()*
  - *НаборЗаписей.Записать() и т.п.*

- **Явные**

- Управляет разработчик
- Методы
  - *НачатьТранзакцию()*
  - *ЗафиксироватьТранзакцию()*
  - *ОтменитьТранзакцию()*

# Важная мысль № 1

**Любое изменение данных (запись, удаление) выполняется в транзакции.**


## Важная мысль № 2

Пока транзакция не зафиксирована (не закончена) можно выполнить «откат транзакции», т.е. отмену всех действий этой транзакции (в том числе уже выполненных)

# **СОБЫТИЯ ЗАПИСИ И ПРОВЕДЕНИЯ ДОКУМЕНТА**

# Запись документа программно

# Запись документа программно




# Запись документа программно

```
Документ ПоступлениеТоваров: Модуль объекта  
  
//1  
⊕ Процедура ПередЗаписью(Отказ, РежимЗаписи, РежимПроведения) ...  
  
//2  
⊕ Процедура ПриЗаписи(Отказ) ...
```

# Проведение документа программно

# Проведение документа программно




# Проведение документа программно


```
Документ Поступление Товаров: Модуль объекта  
  
//1  
⊕ Процедура ПередЗаписью (Отказ, РежимЗаписи, РежимПроведения) ...  
    ↓  
//2  
⊕ Процедура ПриЗаписи (Отказ) ...  
    ↓  
//3  
⊕ Процедура ОбработкаПроведения (Отказ, РежимПроведения) ...
```

# Запись документа из формы

# Запись документа из формы




# Запись документа из формы



# Проведение документа из формы

# Проведение документа из формы



## Важная мысль № 3

Если запись/проведение документа выполняется **не из формы** объекта (например, программно), то **события формы не выполняются.**

### Вывод:

- Если проверка нужна при записи данных из формы, то код проверки нужно разместить в обработчике события формы
- Если проверка нужна всегда (как при записи из формы, так и при программной записи), то код проверки нужно разместить в обработчике события объекта

# МЕТОДИКИ КОНТРОЛЯ ОСТАТКОВ

# Методики контроля остатков

- «Старая» методика
  - Проверяем, есть ли остаток товаров в нужном количестве.  
Если есть – списываем, если нет – сообщаем об ошибке.
  - Слово «старая» не означает что она не используется!
- «Новая» методика
  - Списываем необходимые товары не глядя на остатки.
  - Далее проверяем – образовались ли отрицательные остатки по товарам документа. Если да, то нужно откатить проведение документа (откатить транзакцию).

# Методики контроля остатков

Событие  
ОбработкаПроверкиЗаполнения()

Событие  
ПередЗаписью()

Запись в базу  
данных

Событие  
ПриЗаписи()

Событие  
ОбработкаПроведения()

Списание  
остатков

**«Новая» методика  
контроля:**  
сначала списываем,  
потом проверяем,  
не появились ли  
«минусы»


**«Старая» методика  
контроля:**  
сначала проверяем  
остатки, потом  
списываем

Начало  
транзакции


Конец  
транзакции

# Про что нужно помнить

Остатки из регистра накопления возвращаются **на начало периода** (не включая границу)



# Момент получения остатков



# Объект «Граница»

- Позволяет получать остаток с точностью до момента времени документа
- Можно указать включать ли граничное значение в расчет остатков
- Т.е. остатки из регистра накопления можно получать:
  - На дату (последняя секунда не включается)
  - На момент времени (не включая документ)
  - На момент времени (включая документ) – для этого используется объект «Граница»


# Пример



# Пример

Период	Регистратор	Вид движения	Товар	Количество
20.10.2019 11:59	Приход товара №1	Приход	Холодильник	5
20.10.2019 12:00	Расход товара №1	Расход	Холодильник	2
20.10.2019 12:00	Расход товара №2	Расход	Холодильник	1

- Остаток на дату 20.10.2019 12:00:00 = **5 шт.**
- Остаток на дату 20.10.2019 12:00:01 = **2 шт.**
- Остаток на границу «20.10.2019 12:00:00, регистратор не указан» (вид границы - **включая**) = 2 шт.
- Остаток на границу «20.10.2019 12:00:00 + Расход товара №1» (вид границы - **включая**) = 3 шт.



**«СТАРАЯ»**  
**МЕТОДИКА**  
**КОНТРОЛЯ ОСТАТКОВ**

# «Старая» методика контроля остатков

## 1. Очищаем старые движения документа

- *чтобы не прочитать старые движения документа, если его дата сдвигается вперед)*

## 2. Запросом **получаем остатки товаров** и данные документа

- *остатки получаем на МоментВремени()*

## 3. В цикле **проверяем достаточность** товаров:

- если товаров **не достаточно**, то устанавливаем признак отмены транзакции: **Отказ = Истина**
- если товаров **достаточно** - формируем движения по регистру

## «Старая» методика контроля остатков

# Шаг 1

```
//1. Удаление старых движений документа  
Движения.ОстаткиТоваров.Записывать = Истина;  
Движения.ОстаткиТоваров.Очистить ();  
Движения.Записать ();
```

# «Старая» методика контроля остатков

## Шаг 2

```
//2. Получение остатков товаров текущего документа
Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|   РеализацияТоваровУслугТовары.Номенклатура КАК Номенклатура,
|   СУММА (РеализацияТоваровУслугТовары.Количество) КАК Количество
|ПОМЕСТИТЬ ВТ_ТоварыДокумента
|ИЗ
|   Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
|ГДЕ
|   РеализацияТоваровУслугТовары.Ссылка = &Ссылка
|
|СТРУППИРОВАТЬ ПО
|   РеализацияТоваровУслугТовары.Номенклатура
|;
|
|////////////////////////////////////
|ВЫБРАТЬ
|   ВТ_ТоварыДокумента.Номенклатура КАК Номенклатура,
|   ВТ_ТоварыДокумента.Количество КАК КоличествоВДокументе,
|   ЕСТЬNULL (ОстаткиТоваровОстатки.КоличествоОстаток, 0) КАК КоличествоОстаток,
|   ПРЕДСТАВЛЕНИЕССЫЛКИ (ВТ_ТоварыДокумента.Номенклатура) КАК НоменклатураПредставление
|ИЗ
|   ВТ_ТоварыДокумента КАК ВТ_ТоварыДокумента
|   ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ОстаткиТоваров.Остатки (
|       &МоментВремени,
|       Склад = &Склад
|       И Номенклатура В
|           (ВЫБРАТЬ
|               ВТ_ТоварыДокумента.Номенклатура
|           ИЗ
|               ВТ_ТоварыДокумента)) КАК ОстаткиТоваровОстатки
|   ПО ВТ_ТоварыДокумента.Номенклатура = ОстаткиТоваровОстатки.Номенклатура";

Запрос.УстановитьПараметр ("МоментВремени", МоментВремени ());
Запрос.УстановитьПараметр ("Склад", Склад);
Запрос.УстановитьПараметр ("Ссылка", Ссылка);

РезультатЗапроса = Запрос.Выполнить ();
```



# «Старая» методика контроля остатков

## Шаг 3

```
//3. Проверка остатков
Выборка = РезультатЗапроса.Выбрать ();

Пока Выборка.Следующий () Цикл

    Дефицит = (Выборка.КоличествоВДокументе - Выборка.КоличествоОстаток);
    Если Дефицит > 0 Тогда

        Отказ = Истина;
        Сообщение = Новый СообщениеПользователю;
        Сообщение.Текст = "Недостаточно товара " + Выборка.НоменклатураПредставление + " в количестве " + Дефицит + " ед.";
        Сообщение.Сообщить ();

    КонецЕсли;

    Если Отказ Тогда
        Продолжить;
    КонецЕсли;

    //добавление записей в набор записей регистра "Остатки товаров"
    Движение = Движения.ОстаткиТоваров.Добавить ();
    Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
    Движение.Период = Дата;
    Движение.Номенклатура = Выборка.Номенклатура;
    Движение.Склад = Склад;
    Движение.Количество = Выборка.КоличествоВДокументе;

КонецЦикла;

//Установка флага "Записывать", чтобы произошла запись набора
Движения.ОстаткиТоваров.Записывать = Истина;
```

**«НОВАЯ»**  
**МЕТОДИКА**  
**КОНТРОЛЯ ОСТАТКОВ**

# «Новая» методика контроля остатков

1. **Получаем** запросом **данные документа**
2. **Списываем товары** не глядя на остатки (запись в регистр)
3. **Получаем запросом отрицательные остатки** из регистра  
(*остатки получаем на границу МоментВремени()*  
*включая*)
4. Если результат запроса не пустой, значит товара недостаточно – устанавливаем признак отмены транзакции  
Отказ = Истина, и **сообщаем** пользователю **каких товаров не хватает**

# «Новая» методика контроля остатков

## Шаг 1

```
//1. Получение данных текущего документа запросом
МенеджерВТ = Новый МенеджерВременныхТаблиц;
Запрос = Новый Запрос;
Запрос.МенеджерВременныхТаблиц = МенеджерВТ;
Запрос.Текст =
    "ВЫБРАТЬ
    | РеализацияТоваровУслугТовары.Номенклатура КАК Номенклатура,
    | СУММА (РеализацияТоваровУслугТовары.Количество) КАК КоличествоВДокументе
    | ПОМЕСТИТЬ ВТ_ТоварыДокумента
    | ИЗ
    | Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    | ГДЕ
    | РеализацияТоваровУслугТовары.Ссылка = &Ссылка
    |
    | СГРУППИРОВАТЬ ПО
    | РеализацияТоваровУслугТовары.Номенклатура
    | ;
    |
    | //////////////////////////////////////
    | ВЫБРАТЬ
    | ВТ_ТоварыДокумента.Номенклатура КАК Номенклатура,
    | ВТ_ТоварыДокумента.КоличествоВДокументе КАК КоличествоВДокументе
    | ИЗ
    | ВТ_ТоварыДокумента КАК ВТ_ТоварыДокумента";

Запрос.УстановитьПараметр ("Ссылка", Ссылка);

РезультатЗапроса = Запрос.Выполнить ();

Выборка = РезультатЗапроса.Выбрать ();
```

# «Новая» методика контроля остатков

## Шаг 2

```
//2. Формирование движения в регистре накопления "Остатки товаров"  
Движения.ОстаткиТоваров.Записывать = Истина;  
Пока Выборка.Следующий() Цикл  
    Движение = Движения.ОстаткиТоваров.Добавить();  
    Движение.ВидДвижения = ВидДвиженияНакопления.Расход;  
    Движение.Период = Дата;  
    Движение.Номенклатура = Выборка.Номенклатура;  
    Движение.Склад = Склад;  
    Движение.Количество = Выборка.КоличествоВДокументе;  
КонецЦикла;  
  
Движения.Записать();
```

# «Новая» методика контроля остатков

## Шаг 3

```
//3. Получение отрицательных остатков из регистра
Запрос.Текст =
    "ВЫБРАТЬ
    | ОстаткиТоваровОстатки.Номенклатура КАК Номенклатура,
    | ОстаткиТоваровОстатки.КоличествоОстаток КАК КоличествоОстаток,
    | ПРЕДСТАВЛЕНИЕССЫЛКИ(ОстаткиТоваровОстатки.Номенклатура) КАК НоменклатураПредставление
    | ИЗ
    | РегистрНакопления.ОстаткиТоваров.Остатки(
    |     &МоментВремени,
    |     Склад = &Склад
    |     И Номенклатура В
    |         (ВЫБРАТЬ
    |             ВТ_ТоварыДокумента.Номенклатура
    |             ИЗ
    |             ВТ_ТоварыДокумента КАК ВТ_ТоварыДокумента)) КАК ОстаткиТоваровОстатки
    | ГДЕ
    |     ОстаткиТоваровОстатки.КоличествоОстаток < 0";

Если Режим = РежимПроведенияДокумента.Оперативный Тогда
    ГраницаОстатков = '00010101';
Иначе
    ГраницаОстатков = Новый Граница(МоментВремени(), ВидГраницы.Включая);
КонецЕсли;

Запрос.УстановитьПараметр("МоментВремени", ГраницаОстатков);
Запрос.УстановитьПараметр("Склад", Склад);

РезультатЗапроса = Запрос.Выполнить();
```

# «Новая» методика контроля остатков

## Шаг4

```
//4. Проверка на отрицательные остатки
Если РезультатЗапроса.Пустой() = Ложь Тогда

    Отказ = Истина;

    Выборка = РезультатЗапроса.Выбрать();

    Пока Выборка.Следующий() Цикл
        Сообщение = Новый СообщениеПользователю;
        Сообщение.Текст = "Недостаточно товара " + Выборка.НоменклатураПредставление
            + " в количестве " + (-Выборка.КоличествоОстаток) + " ед.";
        Сообщение.Сообщить();
    КонецЦикла;

КонецЕсли;
```

# Преимущества контроля по «НОВОЙ» методике

- Не нужно очищать старые движения документа
  - Запись в БД - ресурсоемкая операция
- Запрос, который получает отрицательные остатки проще и быстрее, т.к. обращается только к одной таблице - не нужно делать левое соединение с данными документа. Остатки на текущий момент можно получить максимально быстро.

# Применимость «новой» методики контроля остатков

«Новая» методика контроля остатков **может применяться** если:

- В документе есть все необходимые данные для формирования движений в регистре.

*Например, когда данные об остатках товаров по количеству и данные по себестоимости хранятся в разных регистрах*

«Новая» методика контроля остатков **не может применяться** если:

- В документе нет всех данных для формирования движений (нужно делать дополнительный запрос к регистру)

*Например, когда количество и себестоимость товаров хранятся в одном регистре*

# Оперативное проведение

# Оперативное проведение документа

- Документам, которые проводятся текущей датой присваивается текущее время
- Если 2 документа проводятся одновременно, то платформа разнесет документы по разным секундам
- Документы нельзя будет провести будущей датой
- Признак оперативности проведения передается в процедуру **ОбработкаПроведения()**
- При оперативном проведении можно ускорить чтение остатков за счет того, что период можно не указывать (будут получены текущие остатки - их не нужно рассчитывать)

**ЭТО ВСЁ? 😊**

# Демонстрация проблемы

- Имитация параллельного проведения 2-х документов при использовании «старой» методики контроля остатков
- Отрицательные остатки

# Причина проблемы

- Остатки можно читать параллельно в двух разных транзакциях
- Пока ни одна транзакция не зафиксирована, запрос вернет одинаковый результат

# Как устранить проблему

- Запретить читать остатки параллельно в двух разных транзакциях (заблокировать их на время транзакции)
- Тогда вторая транзакция не сможет прочитать остатки, пока не закончится первая транзакция

# Блокировки

- **Блокировка** - это информация о том, что данный ресурс захвачен «кем-то», для выполнения какого-то действия

**Продолжение  
следует...**

# Конец 25-го занятия

Подписывайтесь на нас в социальных сетях 😊



<https://www.youtube.com/channel/UC124tAuP11GlaGpK3YGHzw>



<https://www.facebook.com/ironskillsby/>



[https://www.instagram.com/ironskills\\_/](https://www.instagram.com/ironskills_/)